

LESSON

7



スカッシュゲームを つく 作ろう



がめんじょう うご ほうほう まな
画面上でモノを動かす方法を学ぼう

にゅうりょく

入力してみよう！

つぎ にゅうりょく
次のプログラムを入力しましょう。


F5
RUN

お じっこう みぎ うご
を押して実行すると、●が右に動きます。


```
10 CLS:CLV ⏎  
20 X=15:Y=10:A=-1:B=-1:P=10 ⏎  
60 @LOOP ⏎  
150 LOCATE X,10:PRINT " " ⏎  
160 X=X+1 ⏎  
170 LOCATE X,10:PRINT "●" ⏎  
180 WAIT 5 ⏎  
190 GOTO @LOOP ⏎
```

⏎ は、エンターキーを押す合図です。

へんすう へんすう
Xは変数です。変数と
すうじ い
は、数字を入れておける
はこ
箱のようなものです。



ふ
Xを1増やす
プログラム。






さいしょ はい ぎょう じっこう
Xには最初に15が入っていて、160行が実行されると16になります。


ぎょう ゴートウー く かえ じっこう く かえ
190行のGOTOによって、プログラムは繰り返し実行され、繰り返し
たびにXが1増えます。

ロケイト ひょうじ いち してい めいれい
LOCATEは、キャラクターを表示する位置を指定する命令です。

へんすう ふ ざひょう いち ひょうじ
変数Xを16、17、18…と増やしながら座標(X,10)の位置に●を表示し
ますから、動いているように見えるというわけです。



●を左右に動かしてみよう



ひだり うご
●を左に動かすには、どうしたらよいでしょうか？

ESC お と F4 LIST お
を押して止めて、を押して
プログラムリストを表示し、考えてみましょう。
プログラムを変更したら、忘れず Enter を
押してから、 F5 RUN じっこう
で実行しましょう。

できた
でしょうか？



ぎょう つぎ か ひだり うご
160行を次のように変えれば、●が左に動くようになります。

```
160 X=X-1
```


また **ESC** を押して止めて、160行を次のように変えてみましょう。

```
160 X=X+A
```

Aもやはり変数で、-1が入っているのので●が左に動きます。

では、Aに1を入れたら、どっちに動くと思いますか？


そう、右です。



もしも、●が画面の左端まで行ったときに、Aに1を入れることができたなら、●が右に進み始めます。


次のプログラムを追加してみましょう。●が画面の左端で跳ね返ったら成功です！

```
110 IF X+A=0 A=1
```



●が画面の右端31の位置まで行った時に、左に跳ね返るには、どうしたらよいでしょうか？
110行とほとんど同じ、120行を作ってみましょう。
う。の部分は、何が入るでしょうか？

```
120 IF X+A=31  
A=?
```



さゆう は かえ おと な
左右で跳ね返るときに、音が鳴るようにしてみましょう。

ぎょう ぎょう つぎ つ くわ
110行と120行に、次のプログラムを付け加え

てください。



お わす
を押すのを忘れないように！

```
:BEEP 40,3
```

ここまでの
プログラム



がめん はし さゆう は かえ おと な
●が画面の端で左右に跳ね返り、音が鳴る

```
10 CLS:CLY  
20 X=15:Y=10:A=-1:B=-1:P=10  
  
60 @LOOP  
  
110 IF X+A=0 A=1:BEEP 40,3  
120 IF X+A=31 A=-1:BEEP 40,3  
  
150 LOCATE X,10:PRINT " "  
160 X=X+A  
170 LOCATE X,10:PRINT "●"  
  
180 WAIT 5  
190 GOTO @LOOP
```

じょうげ うご
●が上下に動くようにしましょう

つぎ か こんど うえ うご
プログラムを次のように変えましょう。今度は、●が上に動くようになります。

```
150 LOCATE 15,Y:PRINT " "  
160 Y=Y+B  
170 LOCATE 15,Y:PRINT "●"
```

つぎ にゆうりよく うえ した は かえ
そこで、次のプログラムを入力して、上と下でも跳ね返るようにしてみましょう。

```
130 IF Y+B=0 B=1:BEEP 40,3  
140 IF Y+B=22 B=-1:BEEP 40,3
```

なな うご
●が斜めに動くようにするには、どうしたらよいでしょうか？



ここまでの
プログラム




なな うご がめん はし は かえ
●が斜めに動き、画面の端で跳ね返り、
おと な
音が鳴る。

```
10 CLS:CLV  
20 X=15:Y=10:A=-1:B=-1:P=10  
  
60 @LOOP  
  
110 IF X+A=0 A=1:BEEP 40,3  
120 IF X+A=31 A=-1:BEEP 40,3  
130 IF Y+B=0 B=1:BEEP 40,3  
140 IF Y+B=22 B=-1:BEEP 40,3  
  
150 LOCATE X,Y:PRINT " "  
160 X=X+A:Y=Y+B  
170 LOCATE X,Y:PRINT "●"  
  
180 WAIT 5  
190 GOTO @LOOP
```

いくつかのプログラムを追加して、ゲームにしていきましょう。

つぎにゅうりよく
次のプログラムを入力しましょう。


ドロー めいれい がめん うえ さゆう わく ひょうじ
DRAW命令で、画面の上と左右に枠を表示します。



```
30 DRAW 1,1,62,1
40 DRAW 1,1,1,43
50 DRAW 62,1,62,43
```

つぎひょうじ
次に、パドルを表示しましょう。

← → のカーソルキーで、左右に動かします。



```
70 P=P-BTN(28)+BTN(29)
80 IF P=-1 P=0
90 IF P=25 P=24
100 LOCATE P,22:PRINT "====";
```

がめんした は かえ う かえ
画面下で跳ね返るのではなく、パドルで打ち返せるようにしましょう。

```
140 IF SCR(X+A,Y+B)=61 B=-1:BEEP
15,5:S=S+1
```

ぎょう つか
190行にプログラムを追加します。

う かえ しゅうりよう
●を打ち返せなかったらプログラムが終了します。

```
190 IF Y<22 GOTO @LOOP
```

つぎ しょうり ついか おと な とくてん ひょうじ
次のように、ゲームオーバーの処理を追加しましょう。ブーという音が鳴り、得点が表示されます。ス
いちど はじ
ペースキーでもう一度ゲームを始めることができるようになります。

```
200 BEEP60,30 ☑  
210 LOCATE 5,10:PRINT"SCORE:";S ☑  
220 LOCATE 5,12:PRINT"HIT SPACE  
TO TRY AGAIN!" ☑  
230 IF BTN(32) RUN ELSE CONT ☑
```




さいご ウェイト いどう はや
最後に、WAITの5を3にして、●の移動を速くしま
しょう。

```
180 WAIT 3 ☑
```

2や1にすると、どうなるでしょうか？



はや
遅くなるの？
おそ
遅くなるの？



できあがり



かんせい

完成したプログラム。

さいこうとくてん ねら あそ
最高得点を狙って遊んでみましょう！

```
10 CLS:CLV☑  
20 X=15:Y=10:A=-1:B=-1:P=10☑  
30 DRAW 1,1,62,1☑  
40 DRAW 1,1,1,43☑  
50 DRAW 62,1,62,43☑  
60 @LOOP☑  
70 P=P-BTN(28)+BTN(29)☑  
80 IF P=-1 P=0☑  
90 IF P=25 P=24☑  
100 LOCATE P,22:PRINT"==== ";☑  
110 IF X+A=0 A=1:BEEP 40,3☑  
120 IF X+A=31 A=-1:BEEP 40,3☑  
130 IF Y+B=0 B=1:BEEP 40,3☑  
140 IF SCR(X+A,Y+B)=61B=-1:BEEP1  
5,5:S=S+1☑  
150 LOCATE X,Y:PRINT" "☑  
160 X=X+A:Y=Y+B☑  
170 LOCATE X,Y:PRINT"●"☑  
180 WAIT 3☑  
190 IF Y<22 GOTO @LOOP☑  
200 BEEP60,30☑  
210 LOCATE 5,10:PRINT"SCORE:";S☑  
220 LOCATE 5,12:PRINT"HIT SP☑  
ACE TO TRY AGAIN!"☑  
230 IF BTN(32) RUN ELSE CONT☑
```

遊
ん
で
み
て
ね

